

Sugestões de respostas dos exercícios do Roteiro de Laboratório 6 (Timers)

; Exemplo 1 - sugestão 1

```
RESET      EQU 0H
LTMR0      EQU 0BH          ; local do tratador
STATE      EQU 20H

          ORG RESET          ; PC=0 depois de reset
          JMP INICIO

          ORG LTMR0
          MOV TH0, #0FFH
          MOV TL0, #09CH
          MOV STATE, #1H
          RETI

INICIO:    MOV IE, #10000010B ; habilita TMR0
          MOV TMOD, #01H      ; modo 1
          MOV TH0, #0FFH      ; fundo de escala - 100 = (2^16) - 100
          MOV TL0, #09CH      ; 65.536 - 100 = 65.436 = 0FF9CH
          MOV STATE, #0H      ; inicialização
          MOV R0, #STATE
          MOV DPTR, #TABELA
          MOV R1, #0
          SETB TR0
VOLTA:     CJNE @R0, #1, VOLTA
          MOV STATE, #0H
          MOV A, R1
          MOVC A, @A+DPTR
          MOV P1, A
          INC R1
          CJNE R1, #16, VOLTA
          CLR TR0
          JMP $

TABELA:    DB "MICROCONTROLADOR"
          END
```

; Exemplo 1 - sugestão 2

```

        ORG 0H                      ; PC=0 depois de reset
        JMP INICIO

        ORG 0BH                     ; local do tratador TMR0
        MOV TH0, #0FFH
        MOV TL0, #09CH
        SETB F0
        RETI

INICIO:  MOV IE, #10000010B          ; habilita TMR0
        MOV TMOD, #01H              ; modo 1
        MOV TH0, #0FFH              ; fundo de escala - 100 = (2^16) - 100
        MOV TL0, #09CH              ; 65.536 - 100 = 65.436 = 0FF9CH
        CLR F0                      ; bit 5 do registrador PSW (PSW.5 = sinalização definida pelo programador)
        MOV DPTR, #TABELA
        MOV R1, #0
        SETB TR0
VOLTA:   JNB F0, VOLTA
        CLR F0
        MOV A, R1
        MOVC A, @A+DPTR
        MOV P1, A
        INC R1
        CJNE R1, #16, VOLTA
        CLR TR0
        JMP $

TABELA:  DB "MICROCONTROLADOR"
        END

```

; Exemplo 1 - sugestão 3

```

        ORG 0H                      ; PC=0 depois de reset
        JMP INICIO

        ORG 0BH                     ; local do tratador TMR0
        MOV TH0, #0FFH
        MOV TL0, #09CH
        SETB PSW.1
        RETI

INICIO:  MOV IE, #10000010B          ; habilita TMR0

```

```

        MOV TMOD, #01H           ; modo 1
        MOV TH0, #0FFH           ; fundo de escala - 100 = (2^16) - 100
        MOV TL0, #09CH           ; 65.536 - 100 = 65.436 = 0FF9CH
        CLR PSW.1                 ; bit 1 do registrador PSW = sinalização definida pelo programador
        MOV DPTR, #TABELA
        MOV R1, #0
        SETB TR0
VOLTA:   JNB PSW.1, VOLTA
        CLR PSW.1
        MOV A, R1
        MOVC A, @A+DPTR
        MOV P1, A
        INC R1
        CJNE R1, #16, VOLTA
        CLR TR0
        JMP $

TABELA:  DB "MICROCONTROLADOR"
        END

```

; Exercício 1

```

RESET    EQU 0H
LTMR0    EQU 0BH                 ; local do tratador
STATE    EQU 20H

        ORG RESET                ; PC=0 depois de reset
        JMP INICIO

        ORG LTMR0
        MOV TH0, #0FDH
        MOV TL0, #080H
        MOV STATE, #1H
        RETI

INICIO:  MOV IE, #10000010B       ; habilita TMR0
        MOV TMOD, #01H           ; modo 1
        MOV TH0, #0FDH           ; fundo de escala - 640 = (2^16) - 640
        MOV TL0, #80H            ; 65.536 - 640 = 64.896 = 0FD80H
        MOV STATE, #0H           ; inicialização
        MOV R0, #STATE
        MOV DPTR, #TABELA

```

```

        MOV R1, #0
        SETB TR0
VOLTA:  CJNE @R0, #1, VOLTA
        MOV STATE, #0H
        MOV A, R1
        MOVC A, @A+DPTR
        MOV P1, A
        INC R1
        CJNE R1, #16, VOLTA
        CLR TR0
        JMP $

TABELA: DB "MICROCONTROLADOR"
        END

```

; Exercício 2

```

RESET    EQU 0H
LTMR1    EQU 1BH           ; local do tratador
STATE    EQU 20H

        ORG RESET           ; PC=0 depois de reset
        JMP INICIO

        ORG LTMR1
        MOV TH0, #0ECH
        MOV TL0, #0H
        MOV STATE, #1H
        RETI

INICIO:  MOV IE, #10001000B   ; habilita TMR1
        MOV TMOD, #0H        ; modo 0
        MOV TH1, #0ECH       ; fundo de escala - (640/32) = (2^8)-(640/32) = 256-20 = 236 = 0ECH
        MOV TL1, #0H
        MOV STATE, #0H       ; inicialização
        MOV R0, #STATE
        MOV DPTR, #TABELA
        MOV R1, #0
        SETB TR1
VOLTA:  CJNE @R0, #1, VOLTA
        MOV STATE, #0H
        MOV A, R1

```

```

    MOVC A, @A+DPTR
    MOV P1, A
    INC R1
    CJNE R1, #16, VOLTA
    CLR TR1
    JMP $

```

```

TABELA:  DB "MICROCONTROLADOR"
          END

```

; Exercício 3

```

RESET    EQU 0H
LTMR0    EQU 0BH          ; local do tratador
STATE    EQU 20H

          ORG RESET        ; PC=0 depois de reset
          JMP INICIO

          ORG LTMR0
          MOV STATE, #1H
          RETI

INICIO:   MOV IE, #10000010B    ; habilita TMR0
          MOV TMOD, #00000010B  ; modo 2
          MOV TH0, #80H
          MOV TL0, #80H
          MOV STATE, #0H        ; inicialização
          MOV R0, #STATE
          MOV DPTR, #TABELA
          MOV R1, #0
          MOV R2, #0
          SETB TR0
VOLTA:    CJNE @R0, #1, VOLTA
          MOV STATE, #0H
          INC R2
          CJNE R2, #5, VOLTA
          MOV R2, #0
          MOV A, R1
          MOVC A, @A+DPTR
          MOV P1, A
          INC R1

```

```
CJNE R1, #16, VOLTA
MOV R1, #0
MOV R2, #0
JMP VOLTA
```

```
TABELA:  DB "MICROCONTROLADOR"
          END
```

; Exercício 4

```
RESET     EQU 0H
LTMR0     EQU 0BH
LTMR1     EQU 1BH
STATE     EQU 20H
```

```
ORG RESET           ; PC=0 depois de reset
JMP INICIO
```

```
ORG LTMR0
MOV STATE, #1H
RETI
```

```
ORG LTMR1
CPL P2.3
MOV TH1, #0F1H
RETI
```

```
INICIO:  MOV IE, #10001010B      ; habilita TMR0 e TMR1
         MOV TMOD, #00000010B    ; modo 2 (TMR0) e modo 0 (TMR1)
         MOV TH0, #80H
         MOV TL0, #80H
         MOV TH1, #0F1H
         MOV TL1, #0H
         MOV STATE, #0H          ; inicialização
         MOV R0, #STATE
         MOV R1, #8DH
         MOV DPTR, #TABELA
         MOV R2, #0
         MOV R3, #0
         MOV P2, #0
         SETB TR0
         SETB TR1
```

```

VOLTA:    CJNE @R0, #1, VOLTA
          MOV STATE, #0H
          INC R3
          CJNE R3, #5, VOLTA
          MOV R3, #0
          MOV A, R2
          MOVC A, @A+DPTR
          MOV P1, A
          INC R2
          CJNE R2, #16, VOLTA
          MOV R2, #0
          MOV R3, #0
          JMP VOLTA

TABELA:   DB "MICROCONTROLADOR"
          END

```

; Exercício 5

```

RESET     EQU 0H
LTMR0     EQU 0BH           ; local do tratador TMR0
LEXT1     EQU 13H          ; local do tratador EXT1
STATE     EQU 20H

          ORG RESET         ; PC=0 depois de reset
          JMP INICIO

          ORG LTMR0
          MOV STATE, #1H
          RETI

          ORG LEXT1
          SETB F0
          SETB P3.3
          RETI

INICIO:   MOV IE, #10000110B ; habilita TMR0 e EXT1
          MOV TMOD, #00000010B ; modo 2
          MOV TH0, #80H
          MOV TL0, #80H
          MOV STATE, #0H      ; inicialização
          MOV P3, #08H

```

```

        MOV R0, #STATE
        CLR F0
        MOV DPTR, #TABELA
        MOV R1, #0
        MOV R2, #0
        SETB TR0
VOLTA:   CJNE @R0, #0, TEST_CONT
        JNB F0, VOLTA           ; se F0 = 0 aguarda estouro do TMR0
        CALL NOVO_CONT          ; altera valor da contagem de ciclos de instrução
        JMP VOLTA
TEST_CONT: MOV STATE, #0H
        MOV A, TH0
        CJNE A, #80H, NEXT_CHAR
        INC R2
        CJNE R2, #5, VOLTA
NEXT_CHAR: MOV R2, #0
        MOV A, R1
        MOVC A, @A+DPTR
        MOV P1, A
        INC R1
        CJNE R1, #16, VOLTA
        MOV R1, #0
        MOV R2, #0
        JMP VOLTA

NOVO_CONT: CLR F0
        MOV R7, P2
        MOV A, #0FFH           ; fundo de escala (8 bits)
        CLR C                   ; limpa flag carry
        SUBB A, R7              ; A = A - R7 - C = 0FFH - P2 - 0
        MOV TH0, A
        RET

TABELA:  DB "MICROCONTROLADOR"
        END

```

Sugestões de respostas dos exercícios do Roteiro de Laboratório 7 (Interface Serial)

; Exemplo 1

```

RESET      EQU 00H
LTSERIAL   EQU 23H           ; local tratador

```



```

STATE      EQU 20H

ORG RESET          ; PC=0 depois de reset
JMP INICIO

ORG LTSERIAL
CLR TI
MOV STATE,#1H
RETI

INICIO:  MOV IE, #10010000B
        MOV SCON, #01000000B ; SM0=0 SM1=1 SM2=0 REN=0 TB8=0 RB8=0 TI=0 RI=0
        MOV TMOD, #00100000B ; GATE=0 C/T=0 M1=1 M0=0 GATE=0 C/T=0 M1=0 M0=0 -> modo 2 (carga automática)
        MOV TH1, #0FDH
        MOV TL1, #0FDH
        MOV PCON, #0H
        SETB TR1
        MOV STATE, #0H
        MOV R0, #STATE
        MOV DPTR, #TABELA
        MOV R1, #1
        MOV SBUF, #'M'
VOLTA:   CJNE @R0, #1, VOLTA
        MOV STATE, #0H
        MOV A, R1
        MOVC A, @A+DPTR
        MOV SBUF, A
        INC R1
        CJNE R1, #17, VOLTA
        CLR TR1
        JMP $

TABELA:  DB 'Microcontrolador'
        END

```

; Exercício 1

```

RESET      EQU 00H
LTSERIAL   QUE 23H          ; local tratador

ORG RESET          ; PC=0 depois de reset
JMP INICIO

```

```

        ORG LTSERIAL
        CLR RI
        SETB F0                ; flip-flop de propósito geral do PSW
        RETI

INICIO:  MOV IE, #10010000B
        MOV SCON, #01010000B    ; modo 1 REN=1
        MOV TMOD, #00100000B
        MOV TH1, #0FDH
        MOV TL1, #0FDH
        MOV PCON, #0H
        SETB TR1
VOLTA0:  MOV R1, #30H
VOLTA:   JNB F0, VOLTA
        CLR F0
        MOV @R1, SBUF
        INC R1
        CJNE R1, #36H, VOLTA
        JMP VOLTA0

TABELA:  DB 'Microcontrolador'
        END

```

; Exercício 2

```

RESET    EQU 00H
LTSERIAL EQU 23H                ; local tratador
STATE    EQU 20H

        ORG RESET                ; PC=0 depois de reset
        JMP INICIO

        ORG LTSERIAL
        CLR TI
        MOV STATE, #1H
        RETI

INICIO:  MOV IE, #10010000B
        MOV SCON, #11000000B    ; MODO 3
        MOV TMOD, #00100000B
        MOV TH1, #0FFH

```

```

MOV TL1, #0FFH
MOV PCON, #80H      ; SMOD=1
SETB TR1
MOV STATE, #0H
MOV R0, #STATE
MOV A, #41H
MOV C, P
MOV TB8, C
MOV SBUF, A
VOLTA:  CJNE @R0, #1, VOLTA
MOV STATE, #0H
INC A
MOV C, P
MOV TB8, C
MOV SBUF, A
CJNE A, #61h, VOLTA
MOV A, #40H
JMP VOLTA
END

```

; Exercício 3

```

RESET    EQU 00H
LTSERIAL EQU 23H      ; local tratador
F1        EQU PSW.1    ; FLIP FLOP DE USO DE PROPOSITO GERAL DO PSW  -- F1 NÃO É ACEITO NO KEIL

ORG RESET      ;PC=0 depois de reset
JMP INICIO

ORG LTSERIAL
JNB RI,SETTI
CLR RI
SETB F0        ; FLIP FLOP DE USO DE PROPOSITO GERAL DO PSW
SETTI:  JNB TI,RTN
CLR TI
SETB F1        ; FLIP FLOP DE USO DE PROPOSITO GERAL DO PSW
RTN:      RETI

INICIO:  MOV IE, #10010000B
MOV SCON, #11010000B ; MODO 3; REN=1
MOV TMOD, #00100000B ; TIMER 1 NO MODO 2 - 62.5 kbits/s
MOV TH1, #0FFH

```

```

MOV TL1, #0FFH
MOV PCON, #80H
SETB TR1
MOV A, #41H
MOV C, P
MOV TB8, C
MOV SBUF, A
MOV R1, #20H
VOLTA:  JNB F0, TSTTI
        CLR F0
        MOV @R1, SBUF
        INC R1
        CJNE R1, #26H, TSTTI
        MOV R1, #20H
TSTTI:  JNB F1, VOLTA
        CLR F1
        INC A
        MOV C, P
        MOV TB8, C
        MOV SBUF, A
        CJNE A, #61H, VOLTA
        MOV A, #40H
        JMP VOLTA
        END

```

; Exercício 4

```

RESET    EQU 00H
LTSERIAL EQU 23H
F1        EQU PSW.1

        ORG RESET
        JMP INICIO

        ORG LTSERIAL
        JNB RI,SETTI
        CLR RI
        SETB F0
SETTI:   JNB TI,RTN
        CLR TI
        SETB F1
RTN:     RETI

```

```

INICIO:  MOV IE, #10010000B
        MOV SCON, #01010000B    ; modo 1 (SM0=0 SM1=1) e recepção (REN=1)
        MOV TMOD, #00100000B    ; modo 2
        MOV TH1, #0FDH
        MOV TL1, #0FDH
        MOV PCON, #80H
        MOV R3, #0H
        MOV R5, #41H
        SETB TR1
VOLTA:   MOV SBUF, R5
        JNB F0, TSTTI
        CLR F0
        MOV DPL, R3
        MOV A, SBUF
        MOVX @DPTR, A
        INC R3
        CJNE R3, #0AH, TSTTI
        MOV R3, #0H
TSTTI:   JNB F1, VOLTA
        CLR F1
        INC R5
        MOV SBUF, R5
        CJNE R5, #61H, VOLTA
        MOV R5, #40H
        JMP VOLTA
        END

```

Sugestões de respostas do exercício de revisão para a 2ª prova de laboratório

Enunciado do exercício: Fazer um programa para transmitir pela interface serial a mensagem “Exercício para P2 - EEL7030 _” de forma cíclica a cada 53.748 ciclos de instrução. A transmissão deve ser feita à taxa de 137.5kbits por segundo sem bit de paridade. A contagem dos 53.748 ciclos deve ser feita utilizando o timer 0 (TMR0) no modo temporizador de 16 bits. O programa deve também receber dados, mostrando na porta P1 o valor recebido pela interface serial. Além disso, a quantidade de vezes que a mensagem completa foi transmitida deve ser mostrada na porta P2 sem se preocupar com o fundo de escala da porta P2.

; Sugestão de solução (1)

```
RESET    EQU 0H
```

```

LTMR0    EQU 0BH
LTSERIAL  EQU 23H
STATE    EQU 20H
F1       EQU PSW.1

        ORG RESET
        JMP INICIO

        ORG LTMR0                ; tratador para o timer 0 (0BH)
        CLR TR0
        MOV TH0, #2EH
        MOV TL0, #0CH
        MOV STATE, #1H
        RETI

        ORG LTSERIAL            ; tratador para interface serial (23H)
        JNB RI,SETTI
        CLR RI
        SETB F0
SETTI:   JNB TI,RTN
        CLR TI
        SETB F1
RTN:     RETI

INICIO:  MOV IE, #10010010B      ; EA=1, ES=1 e IT0=1
        MOV SCON, #01010000B    ; modo 1 (SM0=0 SM1=1) e recepção (REN=1)
        MOV TMOD, #00100001B    ; TMR0 (temporizador modo 1) e TMR1 (temporizador modo 2)
        MOV TH1, #1DH
        MOV TL1, #1DH
        MOV TH0, #2EH           ; 65.536d - 53.748d = 11.788d = 2E0Ch
        MOV TL0, #0CH           ; TH0 = 2EH e TL0 = 0CH
        MOV PCON, #0H           ; SMOD = 0
        MOV R0, #20H
        MOV R1, #1H
        MOV R2, #0H
        MOV DPTR, #TABELA
        SETB TR1
        SETB TR0
ESPERA:  CJNE @R0, #1, ESPERA
        MOV STATE, #0H
        MOV SBUF, #'E'
        MOV R1, #1H
VOLTA:   JNB F0, TSTTI
        CLR F0
        MOV P1, SBUF

```

```

TSTTI:    JNB F1, VOLTA
          CLR F1
          MOV A, R1
          MOVC A, @A+DPTR
          MOV SBUF, A
          INC R1
          CJNE R1, #31, VOLTA
          INC R2
          MOV P2, R2
          SETB TR0
          JMP ESPERA

TABELA:   DB 'Exercício para P2 - EEL7030 _ '
          END

```

; Sugestão de solução (2)

```

RESET     EQU 0H
LTMR0     EQU 0BH
LTSERIAL  EQU 23H
STATE     EQU 20H
F1        EQU PSW.1

          ORG RESET
          JMP INICIO

          ORG LTMR0           ; tratador para o timer 0 (0BH)
          CLR TR0
          MOV TH0, #2EH
          MOV TL0, #0CH
          MOV STATE, #1H
          RETI

          ORG LTSERIAL        ; tratador para interface serial (23H)
          JNB RI,SETTI
          CLR RI
          SETB F0
SETTI:    JNB TI,RTN
          CLR TI
          SETB F1
RTN:      RETI

```

```

INICIO:  MOV IE, #10010010B      ; EA=1, ES=1 e IT0=1
        MOV SCON, #01010000B    ; modo 1 (SM0=0 SM1=1) e recepção (REN=1)
        MOV TMOD, #00100001B    ; TMR0 (temporizador modo 1) e TMR1 (temporizador modo 2)
        MOV TH1, #1DH
        MOV TL1, #1DH
        MOV TH0, #2EH           ; 65.536d - 53.748d = 11.788d = 2E0Ch
        MOV TL0, #0CH           ; TH0 = 2EH e TL0 = 0CH
        MOV PCON, #0H           ; SMOD = 0
        MOV R0, #20H
        MOV R1, #1H
        MOV R2, #0H
        MOV DPTR, #TABELA
        SETB TR1
INITX:   MOV SBUF, #'E'
        MOV R1, #1H
VOLTA:   JNB F0, TSTTI
        CLR F0
        MOV P1, SBUF
TSTTI:   JNB F1, VOLTA
        CLR F1
        MOV A, R1
        MOVC A, @A+DPTR
        MOV SBUF, A
        INC R1
        CJNE R1, #31, VOLTA
        INC R2
        MOV P2, R2
        SETB TR0
ESPERA:  CJNE @R0, #1, ESPERA
        MOV STATE, #0H
        JMP INITX

TABELA:  DB 'Exercício para P2 - EEL7030 _ '
        END

```